SM_Kerr

September 22, 2020

1 Kerr spacetime

This notebook demonstrates a few capabilities of SageMath in computations regarding Kerr spacetime. The corresponding tools have been developed within the SageManifolds project.

Click here to download the notebook file (ipynb format). To run it, you must start SageMath within the Jupyter notebook, via the command `sage -n jupyter`

NB: a version of SageMath at least equal to 8.2 is required to run this notebook:

[1]:
```
version()
```

[1]:
```
'SageMath version 9.2.beta13, Release Date: 2020-09-21'
```

First we set up the notebook to display mathematical objects using LaTeX rendering:

[2]:
```
%display latex
```

and we initialize a time counter for benchmarking:

[3]:
```
import time
comput_time0 = time.perf_counter()
```

Since some computations are quite heavy, we ask for running them in parallel on 8 threads:

[4]:
```
Parallelism().set(nproc=8)
```

1.1 Spacetime manifold

We declare the Kerr spacetime (or more precisely the part of it covered by Boyer-Lindquist coordinates) as a 4-dimensional Lorentzian manifold $\mathcal{M}$:

[5]:
```
M = Manifold(4, 'M', latex_name=r'\mathcal{M}', structure='Lorentzian')
print(M)
```

```
4-dimensional Lorentzian manifold M
```

We then introduce the standard Boyer-Lindquist coordinates as a chart `BL` (for *Boyer-Lindquist*) on $\mathcal{M}$, via the method `chart()`, the argument of which is a string (delimited by `r"..."` because of the backslash symbols) expressing the coordinates names, their ranges (the default is $(-\infty, +\infty)$) and their LaTeX symbols:

[6]:
```
BL.<t,r,th,ph> = M.chart(r"t r th:(0,pi):\theta ph:(0,2*pi):\phi")
print(BL); BL
```

```
Chart (M, (t, r, th, ph))
```

[6]:

$$(\mathcal{M}, (t, r, \theta, \phi))$$

[7]:
```
BL[0], BL[1]
```

[7]:

$$(t, r)$$

Metric tensor

The 2 parameters m and a of the Kerr spacetime are declared as symbolic variables:

[8]:
```
var('m, a', domain='real')
```

[8]:

$$(m, a)$$

We get the (yet undefined) spacetime metric:

[9]:
```
g = M.metric()
```

The metric is set by its components in the coordinate frame associated with Boyer-Lindquist coordinates, which is the current manifold's default frame:

[10]:
```
rho2 = r^2 + (a*cos(th))^2
Delta = r^2 -2*m*r + a^2
g[0,0] = -(1-2*m*r/rho2)
g[0,3] = -2*a*m*r*sin(th)^2/rho2
g[1,1], g[2,2] = rho2/Delta, rho2
g[3,3] = (r^2+a^2+2*m*r*(a*sin(th))^2/rho2)*sin(th)^2
g.display()
```

[10]:

$$g = \left(\frac{2\,mr}{a^2\cos(\theta)^2+r^2} - 1\right) \mathrm{d}t \otimes \mathrm{d}t + \left(-\frac{2\,amr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}\right) \mathrm{d}t \otimes \mathrm{d}\phi + \left(\frac{a^2\cos(\theta)^2+r^2}{a^2-2\,mr+r^2}\right) \mathrm{d}r \otimes \mathrm{d}r + \left(a^2\cos(\theta)^2+r^2\right) \mathrm{d}\theta \otimes \mathrm{d}\theta + \left(-\frac{2\,amr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}\right) \mathrm{d}\phi \otimes \mathrm{d}t + \left(\frac{2\,a^2mr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2} + a^2 + r^2\right)\sin(\theta)^2\, \mathrm{d}\phi \otimes \mathrm{d}\phi$$

A matrix view of the components with respect to the manifold's default vector frame:

[11]:
```
g[:]
```

[11]:

$$\begin{pmatrix} \frac{2\,mr}{a^2\cos(\theta)^2+r^2} - 1 & 0 & 0 & -\frac{2\,amr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2} \\ 0 & \frac{a^2\cos(\theta)^2+r^2}{a^2-2\,mr+r^2} & 0 & 0 \\ 0 & 0 & a^2\cos(\theta)^2+r^2 & 0 \\ -\frac{2\,amr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2} & 0 & 0 & \left(\frac{2\,a^2mr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}+a^2+r^2\right)\sin(\theta)^2 \end{pmatrix}$$

The list of the non-vanishing components:

[12]:
```
g.display_comp()
```

[12]:

$$
\begin{array}{lcl}
g_{t\,t} & = & \frac{2\,mr}{a^2\cos(\theta)^2+r^2} - 1 \\
g_{t\,\phi} & = & -\frac{2\,amr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2} \\
g_{r\,r} & = & \frac{a^2\cos(\theta)^2+r^2}{a^2-2\,mr+r^2} \\
g_{\theta\,\theta} & = & a^2\cos(\theta)^2+r^2 \\
g_{\phi\,t} & = & -\frac{2\,amr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2} \\
g_{\phi\,\phi} & = & \left(\frac{2\,a^2mr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}+a^2+r^2\right)\sin(\theta)^2
\end{array}
$$

Levi-Civita Connection

The Levi-Civita connection ∇ associated with g:

[13]:
```
nabla = g.connection()
print(nabla)
```

Levi-Civita connection nabla_g associated with the Lorentzian metric g on the 4-dimensional Lorentzian manifold M

Let us verify that the covariant derivative of g with respect to ∇ vanishes identically:

[14]:
```
nabla(g) == 0
```

[14]: True

Another view of the above property:

[15]:
```
nabla(g).display()
```

[15]: $\nabla_g g = 0$

The nonzero Christoffel symbols (skipping those that can be deduced by symmetry of the last two indices):

[16]:
```
g.christoffel_symbols_display()
```

[16]:

$$\Gamma^{t}{}_{t r} = -\frac{a^{4}m - m r^{4} - \left(a^{4}m + a^{2}m r^{2}\right)\sin(\theta)^{2}}{a^{2}r^{4} - 2\,m r^{5} + r^{6} + \left(a^{6} - 2\,a^{4}m r + a^{4}r^{2}\right)\cos(\theta)^{4} + 2\,\left(a^{4}r^{2} - 2\,a^{2}m r^{3} + a^{2}r^{4}\right)\cos(\theta)^{2}}$$

$$\Gamma^{t}{}_{t \theta} = -\frac{2\,a^{2}m r\cos(\theta)\sin(\theta)}{a^{4}\cos(\theta)^{4} + 2\,a^{2}r^{2}\cos(\theta)^{2} + r^{4}}$$

$$\Gamma^{t}{}_{r \phi} = -\frac{\left(a^{3}m r^{2} + 3\,a m r^{4} - \left(a^{5}m - a^{3}m r^{2}\right)\cos(\theta)^{2}\right)\sin(\theta)^{2}}{a^{2}r^{4} - 2\,m r^{5} + r^{6} + \left(a^{6} - 2\,a^{4}m r + a^{4}r^{2}\right)\cos(\theta)^{4} + 2\,\left(a^{4}r^{2} - 2\,a^{2}m r^{3} + a^{2}r^{4}\right)\cos(\theta)^{2}}$$

$$\Gamma^{t}{}_{\theta \phi} = -\frac{2\,\left(a^{5}m r\cos(\theta)\sin(\theta)^{5} - \left(a^{5}m r + a^{3}m r^{3}\right)\cos(\theta)\sin(\theta)^{3}\right)}{a^{6}\cos(\theta)^{6} + 3\,a^{4}r^{2}\cos(\theta)^{4} + 3\,a^{2}r^{4}\cos(\theta)^{2} + r^{6}}$$

$$\Gamma^{r}{}_{t t} = \frac{a^{2}m r^{2} - 2\,m^{2}r^{3} + m r^{4} - \left(a^{4}m - 2\,a^{2}m^{2}r + a^{2}m r^{2}\right)\cos(\theta)^{2}}{a^{6}\cos(\theta)^{6} + 3\,a^{4}r^{2}\cos(\theta)^{4} + 3\,a^{2}r^{4}\cos(\theta)^{2} + r^{6}}$$

$$\Gamma^{r}{}_{t \phi} = -\frac{\left(a^{3}m r^{2} - 2\,a m^{2}r^{3} + a m r^{4} - \left(a^{5}m - 2\,a^{3}m^{2}r + a^{3}m r^{2}\right)\cos(\theta)^{2}\right)\sin(\theta)^{2}}{a^{6}\cos(\theta)^{6} + 3\,a^{4}r^{2}\cos(\theta)^{4} + 3\,a^{2}r^{4}\cos(\theta)^{2} + r^{6}}$$

$$\Gamma^{r}{}_{r r} = \frac{a^{2}r - m r^{2} + \left(a^{2}m - a^{2}r\right)\cos(\theta)^{2}}{a^{2}r^{2} - 2\,m r^{3} + r^{4} + \left(a^{4} - 2\,a^{2}m r + a^{2}r^{2}\right)\cos(\theta)^{2}}$$

$$\Gamma^{r}{}_{r \theta} = -\frac{a^{2}\cos(\theta)\sin(\theta)}{a^{2}\cos(\theta)^{2} + r^{2}}$$

$$\Gamma^{r}{}_{\theta \theta} = -\frac{a^{2}r - 2\,m r^{2} + r^{3}}{a^{2}\cos(\theta)^{2} + r^{2}}$$

$$\Gamma^{r}{}_{\phi \phi} = \frac{\left(a^{4}m r^{2} - 2\,a^{2}m^{2}r^{3} + a^{2}m r^{4} - \left(a^{6}m - 2\,a^{4}m^{2}r + a^{4}m r^{2}\right)\cos(\theta)^{2}\right)\sin(\theta)^{4} - \left(a^{2}r^{5} - 2\,m r^{6} + r^{7} + \left(a^{6}r - 2\,a^{4}m r^{2} + a^{4}r^{3}\right)\cos(\theta)^{4} + 2\,\left(a^{4}r^{3} - \right.\right.}{a^{6}\cos(\theta)^{6} + 3\,a^{4}r^{2}\cos(\theta)^{4} + 3\,a^{2}r^{4}\cos(\theta)^{2} + r^{6}}$$

$$\Gamma^{\theta}{}_{t t} = -\frac{2\,a^{2}m r\cos(\theta)\sin(\theta)}{a^{6}\cos(\theta)^{6} + 3\,a^{4}r^{2}\cos(\theta)^{4} + 3\,a^{2}r^{4}\cos(\theta)^{2} + r^{6}}$$

$$\Gamma^{\theta}{}_{t \phi} = \frac{2\,\left(a^{3}m r + a m r^{3}\right)\cos(\theta)\sin(\theta)}{a^{6}\cos(\theta)^{6} + 3\,a^{4}r^{2}\cos(\theta)^{4} + 3\,a^{2}r^{4}\cos(\theta)^{2} + r^{6}}$$

$$\Gamma^{\theta}{}_{r r} = \frac{a^{2}\cos(\theta)\sin(\theta)}{a^{2}r^{2} - 2\,m r^{3} + r^{4} + \left(a^{4} - 2\,a^{2}m r + a^{2}r^{2}\right)\cos(\theta)^{2}}$$

$$\Gamma^{\theta}{}_{r \theta} = \frac{r}{a^{2}\cos(\theta)^{2} + r^{2}}$$

$$\Gamma^{\theta}{}_{\theta \theta} = -\frac{a^{2}\cos(\theta)\sin(\theta)}{a^{2}\cos(\theta)^{2} + r^{2}}$$

$$\Gamma^{\theta}{}_{\phi \phi} = -\frac{\left(\left(a^{6} - 2\,a^{4}m r + a^{4}r^{2}\right)\cos(\theta)^{5} + 2\,\left(a^{4}r^{2} - 2\,a^{2}m r^{3} + a^{2}r^{4}\right)\cos(\theta)^{3} + \left(2\,a^{4}m r + 4\,a^{2}m r^{3} + a^{2}r^{4} + r^{6}\right)\cos(\theta)\right)\sin(\theta)}{a^{6}\cos(\theta)^{6} + 3\,a^{4}r^{2}\cos(\theta)^{4} + 3\,a^{2}r^{4}\cos(\theta)^{2} + r^{6}}$$

$$\Gamma^{\phi}{}_{t r} = -\frac{a^{3}m\cos(\theta)^{2} - a m r^{2}}{a^{2}r^{4} - 2\,m r^{5} + r^{6} + \left(a^{6} - 2\,a^{4}m r + a^{4}r^{2}\right)\cos(\theta)^{4} + 2\,\left(a^{4}r^{2} - 2\,a^{2}m r^{3} + a^{2}r^{4}\right)\cos(\theta)^{2}}$$

$$\Gamma^{\phi}{}_{t \theta} = -\frac{2\,a m r\cos(\theta)}{\left(a^{4}\cos(\theta)^{4} + 2\,a^{2}r^{2}\cos(\theta)^{2} + r^{4}\right)\sin(\theta)}$$

$$\Gamma^{\phi}{}_{r \phi} = -\frac{a^{2}m r^{2} + 2\,m r^{4} - r^{5} + \left(a^{4}m - a^{4}r\right)\cos(\theta)^{4} - \left(a^{4}m - a^{2}m r^{2} + 2\,a^{2}r^{3}\right)\cos(\theta)^{2}}{a^{2}r^{4} - 2\,m r^{5} + r^{6} + \left(a^{6} - 2\,a^{4}m r + a^{4}r^{2}\right)\cos(\theta)^{4} + 2\,\left(a^{4}r^{2} - 2\,a^{2}m r^{3} + a^{2}r^{4}\right)\cos(\theta)^{2}}$$

$$\Gamma^{\phi}{}_{\theta \phi} = \frac{a^{4}\cos(\theta)\sin(\theta)^{4} - 2\,\left(a^{4} - a^{2}m r + a^{2}r^{2}\right)\cos(\theta)\sin(\theta)^{2} + \left(a^{4} + 2\,a^{2}r^{2} + r^{4}\right)\cos(\theta)}{\left(a^{4}\cos(\theta)^{4} + 2\,a^{2}r^{2}\cos(\theta)^{2} + r^{4}\right)\sin(\theta)}$$

Killing vectors

The default vector frame on the spacetime manifold is the coordinate basis associated with Boyer-Lindquist coordinates:

[17]:
```
M.default_frame() is BL.frame()
```

[17]: True

[18]:
```
BL.frame()
```

[18]: $\left(\mathcal{M}, \left(\frac{\partial}{\partial t}, \frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi}\right)\right)$

Let us consider the first vector field of this frame:

[19]:
```
xi = BL.frame()[0]; xi
```

[19]: $\frac{\partial}{\partial t}$

[20]:
```
print(xi)
```

```
Vector field d/dt on the 4-dimensional Lorentzian manifold M
```

The 1-form associated to it by metric duality is

[21]:
```
xi_form = xi.down(g)
xi_form.display()
```

[21]:
$$\left(\frac{2\,mr}{a^2\cos(\theta)^2+r^2}-1\right)\mathrm{d}t+\left(-\frac{2\,amr\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}\right)\mathrm{d}\phi$$

Its covariant derivative is

[22]:
```
nab_xi = nabla(xi_form)
print(nab_xi)
nab_xi.display()
```

```
Tensor field of type (0,2) on the 4-dimensional Lorentzian manifold M
```

[22]:
$$\left(\frac{a^2m\cos(\theta)^2-mr^2}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}t\otimes\mathrm{d}r+\left(\frac{2\,a^2mr\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}t\otimes\mathrm{d}\theta+\left(-\frac{a^2m\cos(\theta)^2-mr^2}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}r\otimes\mathrm{d}t+\left(\frac{\left(a^3m\cos(\theta)^2-amr^2\right)\sin(\theta)^2}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}r\otimes\mathrm{d}\phi+\left(-\frac{2\,a^2mr\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}\theta\otimes\mathrm{d}t+\left(\frac{2\left(a^3mr+amr^3\right)\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}\theta\otimes\mathrm{d}\phi+\left(-\frac{\left(a^3m\cos(\theta)^2-amr^2\right)\sin(\theta)^2}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}\phi\otimes\mathrm{d}r+\left(-\frac{2\left(a^3mr+amr^3\right)\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}\phi\otimes\mathrm{d}\theta$$

Let us check that the Killing equation is satisfied:

[23]:
```
nab_xi.symmetrize() == 0
```

[23]: True

Similarly, let us check that $\frac{\partial}{\partial\phi}$ is a Killing vector:

[24]:
```
chi = BL.frame()[3] ; chi
```

[24]: $\frac{\partial}{\partial\phi}$

[25]:
```
nabla(chi.down(g)).symmetrize() == 0
```

[25]: True

Curvature

The Ricci tensor associated with g:

[26]:
```
Ric = g.ricci()
print(Ric)
```

```
Field of symmetric bilinear forms Ric(g) on the 4-dimensional Lorentzian
manifold M
```

Let us check that the Kerr metric is a solution of the vacuum Einstein equation:

[27]:
```
Ric == 0
```

[27]:

True

Another view of the above property:

[28]:
```
Ric.display()
```

[28]:

$\mathrm{Ric}\left(g\right) = 0$

The Riemann curvature tensor associated with g:

[29]:
```
R = g.riemann()
print(R)
```

```
Tensor field Riem(g) of type (1,3) on the 4-dimensional Lorentzian manifold M
```

Contrary to the Ricci tensor, the Riemann tensor does not vanish; for instance, the component $R^0{}_{123}$ is

[30]:
```
R[0,1,2,3]
```

[30]:

$$-\frac{\left(\left(a^{7} m - 2 \, a^{5} m^{2} r + a^{5} m r^{2}\right) \cos\left(\theta\right)^{5} - \left(3 \, a^{7} m - 2 \, a^{5} m^{2} r + 8 \, a^{5} m r^{2} - 6 \, a^{3} m^{2} r^{3} + 5 \, a^{3} m r^{4}\right) \cos\left(\theta\right)^{3} + 3 \, \left(3 \, a^{5} m r^{2} - 2 \, a^{3} m^{2} r^{3} + 5 \, a^{3} m r^{4} + 2 \, a m r^{6}\right) \cos\left(\theta\right)\right) \mathrm{s}}{a^{2} r^{6} - 2 \, m r^{7} + r^{8} + \left(a^{8} - 2 \, a^{6} m r + a^{6} r^{2}\right) \cos\left(\theta\right)^{6} + 3 \, \left(a^{6} r^{2} - 2 \, a^{4} m r^{3} + a^{4} r^{4}\right) \cos\left(\theta\right)^{4} + 3 \, \left(a^{4} r^{4} - 2 \, a^{2} m r^{5} + a^{2} r^{6}\right) \cos\left(\theta\right)^{2}}$$

Bianchi identity

Let us check the Bianchi identity $\nabla_p R^i{}_{jkl} + \nabla_k R^i{}_{jlp} + \nabla_l R^i{}_{jpk} = 0$:

[31]:
```
DR = nabla(R)     #long (takes a while)
print(DR)
```

```
Tensor field nabla_g(Riem(g)) of type (1,4) on the 4-dimensional Lorentzian
manifold M
```

[32]:
```
# from __future__ import print_function  # uncomment for SageMath version < 9.0␣
 ↪(Python 2 based)
for i in M.irange():
    for j in M.irange():
        for k in M.irange():
            for l in M.irange():
                for p in M.irange():
                    print(DR[i,j,k,l,p] + DR[i,j,l,p,k] + DR[i,j,p,k,l], end='␣
 ↪')
```

```
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
```

```
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
```

If the last sign in the Bianchi identity is changed to minus, the identity does no longer hold:

[33]:
```
DR[0,1,2,3,1] + DR[0,1,3,1,2] + DR[0,1,1,2,3] # should be zero (Bianchi identity)
```

[33]: 0

[34]:
```
DR[0,1,2,3,1] + DR[0,1,3,1,2] - DR[0,1,1,2,3] # note the change of the second + to -
```

[34]:
$$\frac{24\left(\left(a^{5}mr+a^{3}mr^{3}\right)\cos(\theta)\sin(\theta)^{3}-\left(a^{5}mr-amr^{5}\right)\cos(\theta)\sin(\theta)\right)}{a^{2}r^{6}-2\,mr^{7}+r^{8}+(a^{8}-2\,a^{6}mr+a^{6}r^{2})\cos(\theta)^{6}+3\,(a^{6}r^{2}-2\,a^{4}mr^{3}+a^{4}r^{4})\cos(\theta)^{4}+3\,(a^{4}r^{4}-2\,a^{2}mr^{5}+a^{2}r^{6})\cos(\theta)^{2}}$$

1.1.1 Kretschmann scalar

The tensor $R^\flat$, of components $R_{abcd} = g_{am} R^m{}_{bcd}$:

[35]:
```
dR = R.down(g)
print(dR)
```

```
Tensor field of type (0,4) on the 4-dimensional Lorentzian manifold M
```

The tensor $R^\sharp$, of components $R^{abcd} = g^{bp} g^{cq} g^{dr} R^a{}_{pqr}$:

[36]:
```
uR = R.up(g)
print(uR)
```

Tensor field of type (4,0) on the 4-dimensional Lorentzian manifold M

The Kretschmann scalar $K := R^{abcd} R_{abcd}$:

[37]:
```
Kr_scalar = uR['^{abcd}']*dR['_{abcd}']
Kr_scalar.display()
```

[37]:

$$\begin{array}{llcl} & \mathcal{M} & \longrightarrow & \mathbb{R} \\ & (t,r,\theta,\phi) & \longmapsto & -\frac{48\left(a^6 m^2 \cos(\theta)^6 - 15\,a^4 m^2 r^2 \cos(\theta)^4 + 15\,a^2 m^2 r^4 \cos(\theta)^2 - m^2 r^6\right)}{a^{12}\cos(\theta)^{12} + 6\,a^{10} r^2 \cos(\theta)^{10} + 15\,a^8 r^4 \cos(\theta)^8 + 20\,a^6 r^6 \cos(\theta)^6 + 15\,a^4 r^8 \cos(\theta)^4 + 6\,a^2 r^{10}\cos(\theta)^2 + r^{12}} \end{array}$$

A variant of this expression can be obtained by invoking the factor() method on the coordinate function representing the scalar field in the manifold's default chart:

[38]:
```
Kr = Kr_scalar.coord_function()
Kr.factor()
```

[38]:

$$-\frac{48\left(a^2\cos(\theta)^2 + 4\,ar\cos(\theta) + r^2\right)\left(a^2\cos(\theta)^2 - 4\,ar\cos(\theta) + r^2\right)(a\cos(\theta) + r)(a\cos(\theta) - r)m^2}{\left(a^2\cos(\theta)^2 + r^2\right)^6}$$

As a check, we can compare Kr to the formula given by R. Conn Henry, Astrophys. J. 535, 350 (2000):

[39]:
```
Kr == 48*m^2*(r^6 - 15*r^4*(a*cos(th))^2 + 15*r^2*(a*cos(th))^4
              - (a*cos(th))^6) / (r^2+(a*cos(th))^2)^6
```

[39]: True

The Schwarzschild value of the Kretschmann scalar is recovered by setting $a = 0$:

[40]:
```
Kr.expr().subs(a=0)
```

[40]: $\frac{48\,m^2}{r^6}$

Let us plot the Kretschmann scalar for $m = 1$ and $a = 0.9$:

[41]:
```
K1 = Kr.expr().subs(m=1, a=0.9)
plot3d(K1, (r, 1, 3), (th, 0, pi), axes_labels=['r', 'theta', 'Kr'])
```

[41]: Graphics3d Object

[42]:
```
print("Total elapsed time: {} s".format(time.perf_counter() - comput_time0))
```

Total elapsed time: 483.1138785999992 s